
(モデル理論アプローチによるシミュレーション)

遺伝アルゴリズムのCAST言語によるモデル
東洋大学　旭貴朗
　　　目次
　１ はじめに

　２ 遺伝アルゴリズム

　３ CAST言語によるオートマトンモデル

　４ おわりに

１ はじめに

　本稿では，遺伝アルゴリズム（Genetic Algorithm : GA）を行なうオートマトンをCAST言語を用いて構成する．オートマトン１回の状態遷移で，遺伝アルゴリズムの１サイクル（１世代）の計算を行なう．本稿では，遺伝子プールの行列を状態とし，状態遷移は行列計算で行なう．
　このサイト「モデル記述言語CASTによるシミュレーション」で公開されているsimcast.zip（＋extProlog）をインストールすると，モデル理論アプローチによるシミュレーション開発実行環境Simcastができあがり，本文中のモデルを実行することができる．開発実行環境の操作方法などは，同サイトや高原康彦ほか『形式手法 モデル理論アプローチ：情報システム開発の基礎』日科技連(2007)（少なくとも第３章3.1項pp.87-103）を参照されたい．

２　遺伝アルゴリズム

図１　遺伝アルゴリズム

　遺伝アルゴリズム（図1）の手順とExcel上のVBA言語によるプログラムは，Web資料 [注１] を参照されたい．本稿では，そのページの「自治体税収の近似式」を求める例題を，CAST言語でモデル化する．

３　CAST言語によるモデル
　近似式を求める遺伝アルゴリズムのCAST言語によるオートマトンモデルを構成したい．上記Webページの図表３「近似式の推定(Excel画面)」を見ると，右側に20個の遺伝子と適応度（fitness)を表示した表（20行×3列の行列）がある．その表中の右2列が遺伝子を表わしている．この20行×2列の行列が遺伝子プールである．これをオートマトンの状態とする．

　つまり，ひとつの個体（直線）の遺伝子は直線の切片と傾きの組[u,v]である．状態はそれら20組をリストにして並べたものc=[ [u1,v1], [u2,v2], .... , [u20,v20] ] である．言語CASTでは，リストのリストは「行列」と考えて処理できる．
　単に状態遷移だけを考えるので，出力は不要である．したがって，正確には状態機械（state machine）モデルを構成する．遺伝アルゴリズムを行なう状態機械のCAST言語によるモデルは図２のようになる

//ga05b.set 遺伝アルゴリズム（defList計算）

func([y,f,fit,rand,h,g]);

  
y(1)=0.602; y(2)=1.346; y(3)=1.247; y(4)=2.350; y(5)=2.597;

  
f(u,v,x)=abs(u+v*x - y(x));

  
fit(u,v)=f(u,v,1)+f(u,v,2)+f(u,v,3)+f(u,v,4)+f(u,v,5);

  
rand(n,m)=r <-> myrandom(r1), r:=n+(m-n)*r1;

  
h()=rand(-5,5)*pow(0.1,rand(1,10));

  
g(L,i,j)=M <-> 



[[u1,v1],[u2,v2]]:=project(L,[i,j]), 



M:=[[u1+h(),v2+h()],[u2+h(),v1+h()]]; //交叉+突然変異

inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=defList(p1(z,x,[]),member(x,genIndex(1,20)));

delta(c,a)=cc <-> 


//世代の開始


Ws:=defList(p2(z,[u,v],[]),member([u,v],c)), //適応度計算


c1:=sort(Ws), //適応度順の個体リスト


Ps:=projectMatrix(c1,1,2,10,3), //親を選択


L:=[g(Ps,1,2),g(Ps,3,4),g(Ps,5,6),g(Ps,7,8),g(Ps,9,10)],


Qs:=append(L), //子を確定


cc:=append(Ps,Qs); //次世代が確定

p1(z,x,[]) <-> z:=[rand(-10,10),rand(-10,10)];

p2(z,[u,v],[]) <-> z:=[fit(u,v),u,v];
　　　図２　遺伝アルゴリズムを行なう状態機械のCAST言語によるモデル
　図２の初期状態initialstateの定義では一様乱数を発生させて20個の初期遺伝子を設定している．つまり，状態cは，各行が個体の遺伝子であり，それが20行並んだ行列である．

　状態遷移delta(c,a)=ccは次のように計算される．まず適応度fit(u,v)を計算し，それを付加した個体[fit(u,v),u,v]のリストWsを作成する．これをソートすれば，適応度順に並んだ個体リストc1ができる．

　次のPsは行列c1の第1行第2列から第10行第3列までを抜き出しており，適応度の高い親グループを作成している．関数g(Ps,1,2)は，適応度が1位の親と2位の親から，遺伝子交叉により２子を作成するものである．そのとき（小さな数字の乱数h()を加えて）突然変異も行なっている．ちなみに，rand(n,m)は数値nからmまでの間で乱数を取得する関数である．したがって，Qsは子グループである．ここではWebページと同じように親グループと子グループを合わせて次世代ccとしている．
　入力系列の定義inputsequence()="d"; times()=1000;ではダミーの"d"をaに1000回代入して時間を進める（1000世代の進化を計算する）．シミュレーション開発実行環境Simcastで図2のモデルをコンパイルし実行した結果を図３に示す[注2]．実行時間は4秒であった．

図３　実行結果

　また，図２では遺伝アルゴリズムの計算部分だけを示してあるが，データのグラフ表示やスプレッドシート表示の機能はない．文献[注3,4]を参考にして，データ表示とファイル保存も加えると付録のようになる．実行時間は18秒であった．
４ おわりに

　本稿では，参考文献と同じ遺伝アルゴリズムを別の言語CASTで書き直し，実行した．実行結果は参考文献の結果と一致している．しかしながら，手続型言語とはまったく異なる形式のユーザモデルになっている．

　CAST言語によるシミュレーション開発のポイントは，何をオートマトンの状態にするかである．今回は，遺伝子プールの行列（20行×2列）を状態としたので，ベクトル計算と転置行列をとることにより，見た目は簡単に（文字数は少なく）記述できている．
　モデルの開発は，関数delta(c,a)の定義を１行作成するごとにコンパイル・実行し，確認しながら行なえる．例えば，まず，

Ws:=defList(p2(z,[u,v],[]),member([u,v],c)),
cc:=Ws;
p2(z,[u,v],[]) <-> z:=[fit(u,v),u,v];
で実行し確認する．次にc1:=sort(Ws)を加えて再実行するなどである．

　したがって，行列（リストのリスト）の処理に慣れないユーザにとっては，行列がどのように変化していくかを体験しながら開発することができる．もちろん行列処理に慣れたユーザなら，（頭の中の）行列イメージがモデルのコードに正確に対応しているのでモデリングは容易だろう．

注と参考文献

[1] 旭貴朗「Excelによる進化シミュレーション」　(2008.03.29)

　http://www2.toyo.ac.jp/~asahi/education/soron/shiryo/programing/language/excel/genetic/ 

[2] 出力表示は，既存の形式では読みにくいので，オートマトン実行エンジンを書き直してチューニングした．図３のように行列C2を行列らしく表示する．

[3] 旭貴朗「グラフによるデータ表示」技術資料

　http://www2.toyo.ac.jp/~asahi/research/simulation/docs/graph.doc　(2008.04.11)
[4] 旭貴朗「スプレッドシートによるデータ表示」技術資料

　http://www2.toyo.ac.jp/~asahi/research/simulation/docs/spreadsheet.doc　(2008.04.14)
付録（グラフ表示，シート表示，ファイル保存を追加した場合）

//ga11.set 遺伝アルゴリズム（全データ表示）
func([y,f,fit,rand,h,g]);

  
y(1)=0.602; y(2)=1.346; y(3)=1.247; y(4)=2.350; y(5)=2.597;

  
f(u,v,x)=abs(u+v*x - y(x));

  
fit(u,v)=f(u,v,1)+f(u,v,2)+f(u,v,3)+f(u,v,4)+f(u,v,5);

  
rand(n,m)=r <-> myrandom(r1), r:=n+(m-n)*r1;

  
h()=rand(-5,5)*pow(0.1,rand(1,10));

  
g(L,i,j)=M <-> 



[[u1,v1],[u2,v2]]:=project(L,[i,j]), 



M:=[[u1+h(),v2+h()],[u2+h(),v1+h()]];

preprocess() <->


getWp("sheet",Wp), //スプレッドシート表示


(Wp >0) -> (Wp.g:=Wp)


otherwise (



L:=defList(p1(z,x,[]),member(x,genIndex(1,20))),



makesheet(20,3,Wp1), Wp.g:=Wp1,



showGeneration(L) );
inputsequence()="d";

times()=1000;

initialstate()=c0 <-> c0:=projectss(Wp.g,1,2,20,3);

delta(c,a)=cc <-> 


//世代の開始


Ws:=defList(p2(z,[u,v],[]),member([u,v],c)), //適応度計算


c1:=sort(Ws), //適応度順の個体リスト


showGeneration(c1),


Ps:=projectMatrix(c1,1,2,10,3), //親を選択


xwriteln(0,"Ps="),xwriteln(0,Ps),


L:=[g(Ps,1,2),g(Ps,3,4),g(Ps,5,6),g(Ps,7,8),g(Ps,9,10)],


Qs:=append(L), //子を確定


cc:=append(Ps,Qs); //次世代が確定

p1(z,x,[]) <-> z:=[0, rand(-10,10), rand(-10,10)];

p2(z,[u,v],[]) <-> z:=[fit(u,v),u,v];

showGeneration(c1) <-> 


appendf("ga11.lib","w", c1), //ファイルに書き込み


writess(Wp.g,"r",1,1,c1),  //スプレッドシートに書き込み


[[w,u,v],[z,x,y]]:=project(c1,[1,10]),


d1:=[y(1),y(2),y(3),y(4),y(5)],


d2:=[u+v*1, u+v*2, u+v*3, u+v*4, u+v*5],


d3:=[x+y*1, x+y*2, x+y*3, x+y*4, x+y*5],


data:=[d1,d2,d3],


show1(data,plot); //グラフに表示

